
Key to Midterm Exam S3
Computer Architecture

Duration: 1 hr. 30 min.

Exercise     1 (5 points)
Complete the table shown on the  answer sheet. Write down the new values of the registers (except the
PC) and memory that are modified by the instructions. Use the hexadecimal representation. Memory
and registers are reset to their initial values for each instruction.

Initial values: D0 = $0004FFFF  A0 = $00005000  PC = $00006000

D1 = $FFFF0005  A1 = $00005008
D2 = $FFFFFFFE  A2 = $00005010

$005000  54 AF 18 B9 E7 21 48 C0
$005008  C9 10 11 C8 D4 36 1F 88
$005010  13 79 01 80 42 1A 2D 49

Exercise     2 (4 points)
Complete the table shown on the answer sheet. Give the result of the additions and the values of the N, Z,
V and C flags.

Exercise     3 (3 points)
Write a few instructions that modify D1 so that it takes the values given on the answer sheet. For each
case, the initial value of D1 is $76543210. Use ROR, ROL or SWAP only. Answer on the  answer sheet.

Exercise     4 (2 points)
Answer the questions on the answer sheet.

Exercise     5 (6 points)
Let us consider the following program:

Main        move.l  #$23456789,d7 ; $23456789 -> D7.L

next1       moveq.l #1,d1         ; $00000001 -> D1.L
            tst.b   d7            ; Set N and Z according to D7.B.
            bmi     next2         ; Branch if N = 1 (D7.B < 0).
            moveq.l #2,d1         ; Otherwise, $00000002 -> D1.L

next2       moveq.l #1,d2         ; $00000001 -> D2.L
            tst.w   d7            ; Set N and Z according to D7.W.
            bpl     next3         ; Branch if N = 0 (D7.W ≥ 0).
            moveq.l #2,d2         ; Otherwise, $00000002 -> D2.L

next3       clr.l   d3            ; $00000000 -> D3.L
            move.w  #$4321,d0     ; $4321 -> D0.W (D0.B = $21)
loop3       addq.l  #1,d3         ; D3.L + 1 -> D3.L
            subq.b  #1,d0         ; D0.B – 1 -> D0.B ; Only D0.B is decremented.
            bne     loop3         ; Branch if Z = 0 (D0.B ≠ 0)

next4       clr.l   d4            ; $00000000 -> D4.L
            move.w  #$44,d0       ; $0044 -> D0.W
loop4       addq.l  #1,d4         ; D4.L + 1 -> D4.L
            dbra    d0,loop4      ; DBRA = DBF ; D0.W – 1 -> D0.W
                                  ; Branch if D0.W ≠ -1 (D0.W ≠ $FFFF)

next5       clr.l   d5            ; $00000000 -> D5.L
            moveq.l #10,d0        ; $0000000A -> D0.L
loop5       addq.l  #1,d5         ; D5.L + 1 -> D5.L
            addq.l  #1,d0         ; D0.L + 1 -> D0.L
            cmpi.l  #30,d0        ; Compare D0.L to 30.
            bne     loop5         ; Branch if Z = 0 (D0.L ≠ 30)

next6       moveq.l #1,d6         ; $00000001 -> D6.L
            cmp.b   #$70,d7       ; Compare D7.B to $70.
            blt     quit          ; Branch if D7.B < $70 (signed comparison).
            moveq.l #2,d6         ; Otherwise, $00000002 -> D6.L

quit        illegal

Complete the table shown on the answer sheet.

Last name: ............................................. First name: ........................................... Group: ............................ 

ANSWER SHEET TO BE HANDED IN

Exercise     1

Instruction Memory Register

Example $005000  54 AF 00 40 E7 21 48 C0
A0 = $00005004
A1 = $0000500C

Example $005008  C9 10 11 C8 D4 36 FF 88 No change

MOVE.L  (A2)+,(A0)+ $005000  13 79 01 80 E7 21 48 C0
A0 = $00005004
A2 = $00005014

MOVE.L  4(A2),4(A0) $005000  54 AF 18 B9 42 1A 2D 49 No change

MOVE.B  $500A,-1(A1,D0.W) $005000  54 AF 18 B9 E7 21 11 C0 No change

MOVE.L  #$500A,-5(A1,D1.W) $005008  00 00 50 0A D4 36 1F 88 No change

MOVE.W  $500A,-(A1) $005000  54 AF 18 B9 E7 21 11 C8 A1 = $00005006

Exercise     2

Operation
Size
(bits)

Result
(hexadecimal)

N Z V C

$F0 + $11 8 $01 0 0 0 1

$F0 + $11 16 $0101 0 0 0 0

$8000 + $8000 16 $0000 0 1 1 1

$40000000 + $80000000 32 $C0000000 1 0 0 0

Exercise     3
Final value of D1 : $76542301. Use four lines of instructions at the most.

                                    ; D1 = $ 7654 3210
            ror.b   #4,d1           ; D1 = $ 7654 3201
            ror.w   #8,d1           ; D1 = $ 7654 0132
            ror.b   #4,d1           ; D1 = $ 7654 0123
            ror.w   #8,d1           ; D1 = $ 7654 2301

Final value of D1 : $54231067. Use four lines of instructions at the most.

                                    ; D1 = $ 7654 3210
            ror.l   #8,d1           ; D1 = $ 1076 5432
            ror.b   #4,d1           ; D1 = $ 1076 5423
            swap    d1              ; D1 = $ 5423 1076
            ror.b   #4,d1           ; D1 = $ 5423 1067

Exercise     4

Question Answer

Give two assembler directives. ORG, DC

How many status register does the 68000 have? Only one

What is the size of the CCR register? 8 bits

Which 68000 mode has limited privileges? The user mode

Exercise     5

Values of registers after the execution of the program.
Use the 32-bit hexadecimal representation.

D1 = $00000001 D4 = $00000045

D2 = $00000001 D5 = $00000014

D3 = $00000021 D6 = $00000001

